Behavior Research Methods & Instrumentation
1978, Vol. 10 (1), 91-94

A program to generate synthetic walkers as dynamic point-light displays

JAMES E. CUTTING
Wesleyan University, Middletown, Connecticut 06457

Using a technique that began with the work of Marey (1895/1972), Johansson (1973, 1975, 1976) has begun to explore the underlying stimulus structure of biological motion. One version of the technique attaches glass-bead retroreflectant tape to the joints of human beings, focuses bright lights on them, and video records their movements. The contrast of the image is turned up and the brightness down so that only a system of moving point-lights is seen. The technique is important because it deprives the stimulus of familiarity cues that could be used for recognition, but does little to the potency of the percept. The viewer "sees" the formless invariant relations of the display, which are readily interpretable as a human figure in motion.

My colleagues and I have begun to explore various attributes of gait perception (Barclay, Cutting, & Kozlowski, 1978; Cutting & Kozlowski, 1977; Kozlowski & Cutting, 1977). It then occurred to me that the study of gait perception could be advanced through the simultaneous study of gait synthesis. This general two-pronged attack has proved invaluable in speech perception (Cooper, Liberman, & Borst, 1951; Liberman, Cooper, Shankweiler, & Studdert-Kennedy, 1967).

Hardware and Software. The program listed in Appendix A represents the culmination of many efforts in measuring natural gait and converting those measurements into algorithms appropriate for gait synthesis. It was written for a Tektronix 604 monitor display scope, driven by a Data General Nova. It fits easily within the 32K capacity of that machine. The Tektronix scope has a display capacity of 1,024 by 1,024 points, but the program is easily modifiable for other scope sizes and grains. Subroutine supports for the program include those from the FORTRAN and MEGATEK libraries.

The computational core of the program is a DO LOOP that increments around the 360 deg of a circle. At many points within that circle, many sine and cosine computations are made to determine the particular locales of 13 points. These points divide into three groups according to their visibility in a body walking from left to right, yielding saggital projections to the viewer. Those points seen at all times correspond to the head and the right shoulder, elbow, wrist, knee, and ankle joints. Those points seen intermittently are the right hip (occluded by the right arm once every stride), the left elbow and wrist (occluded by the body once every stride), the left knee (occluded by the right leg once every stride), and the left ankle (occluded once every other stride by the right foot). Two points, the left shoulder and hip, are never displayed. The movements of the shoulders and hips are ellipsoidal (see Carlsöö, 1972), and the movements of the arms and legs are pendular. The entire body moves laterally across the screen according to step size, and up and down in a slight bouncing motion.

Satisfactory displays can be obtained by incrementing around the circle in 9-deg steps, yielding 40 frames per step cycle (a group of two steps). A natural-looking gait can be obtained by displaying each frame an average of 30 msec, with a range of 15 to 45 msec varying according to the walker's position within the step cycle. All 120 frames of a walker display are generated by the Nova within about 12 sec. Thus, it is possible to generate a test sequence of many trials and many different stimuli through on-line interaction with the computer.

Availability. Requests for additional information or reprints should be addressed to the author care of Wesleyan University, Middletown, Connecticut 06457.

APPENDIX A

```
COMMON /INPUT/ K1, K2, K3
DIMENSION SI(0:810), SIA(0:540)                              ;FOR SIN, COS, & ANKLE TABLES
DIMENSION IA(2880)                                           ;FOR DISPLAY TABLE
REAL LEG
EXTERNAL PIDCT
CALL FINTD(35,PIDCT)
TORSO=85                    ;SCREEN UNITS, GOOD SIZE TO FIT
ACCEPT "STEP SIZE IN PROPORTION TO HEIGHT (ABOUT .40) =",STEP
STEP=STEP*TORSO*3.25        ;STEP AFFECTS HIP & SHOULDER MOVEMENTS
ACCEPT "HIP SWING VALUE, X-AXIS EXCURSION (M<1,F>1) =",HEX
ACCEPT "SHOULDER EXCURSION AS MULTIPLE OF HIP (M>3,F<1) =",SHER
ACCEPT "TIME (MODAL IS 300, MIN 150) =", TIME                ;CMSEC PER FRAME
                                                             ;AFFECTS LEAN & DELAY OF ANKLE
K=12                        ;DISPLAY BRIGHT LIGHTS
IP=0                        ;INITIALIZE DISPLAY TABLE POINTER
CALL DINIT(IA(1),2880,IP)                                    OPEN MEMORY FOR DISPLAY TABLE
IST=IP                      ;STORE POINTER FOR BEGINNING OF DISPLAY TABLE
```

```
      HEX=HEX*.03*STEP              ;HIP ELLIPSE, X AXIS
      HEY=HEX/4.                    ;HIP ELLIPSE, Y AXIS
      SEX=SHER*HEX                  ;SHOULDER ELLIPSE, X AXIS
      SEY=SEX/5.                    ;SHOULDER ELLIPSE, Y AXIS
      FEMUR=.77*TORSO               ;LENGTH OF UPPER LEG
      TIBIA=.77*TORSO               ;LENGTH OF LOWER LEG
      LEG=FEMUR+TIBIA+.3*TORSO        ;ADJUST FOR ANKLE-TO-HEEL LENGTH
      HUMOR=.59*TORSO               ;LENGTH OF UPPER ARM
      ULNA=.56*TORSO                ;LENGTH OF LOWER ARM
      HSTEP=STEP/2.-HEX             ;HALFSTEP WITHOUT TORSO TORQUE
      BOUNC=(LEG-SQRT(LEG*LEG-HSTEP*HSTEP)-HEY)/2.  ;UP & DOWN MOVEMENT
                                          ;OF BODY DUE TO STEP SIZE & HIP ROLL
      RAD=.017453                   ;CONSTANT FOR DEGREE-TO-RADIAN CONVERSION
      SWF=ATAN(HSTEP/SQRT(LEG*LEG-HSTEP*HSTEP))  ;FEMUR SWING
      SWH=.24*(STEP-4.*SEX)*RAD             ;HUMOR SWING, F(SHOULDER SWING & STEP)
      SWU=1.65*SWH                  ;ULNA SWING
      SWT=25.*RAD+SWF+5.*HEY*RAD                  ;TIBIA SWING, 25=ATAN(FOOT/TIBIA)
                                                  ;5.*HEY IS INCREASE DUE TO HIP ROLL
      FEMFO=RAD*.04*STEP            ;FORWARD TILT OF FEMUR PENDULUM
      COSFF=LEG*(COS(SWF-FEMFO)-COS(SWF+FEMFO))/2           ;Y CORRECTION FOR HIP
C
      DO 10 I=0,809                 ;ESTABLISH SIN TABLE, 2.25 CIRCUITS
10    SI(I)=SIN(I*RAD)
C
      DELAY=10000./(TIME-50)        ;DEGREES PRE & POST FEMUR SWING THAT
                                    ;TIBIA SWINGS, DUE TO FOOT AND TO COMPOUND PENDULUM
      DO 30 I=0,180                 ;ANKLE TABLE, BROAD W-SHAPE, 0-540 DEGREES
      SIA(I)=0.
30    SIA(I+360)=0.
      N=360.+DELAY
      I=0
      DO 35 J=270,N                 ;SYMMETRIC ABOUT 270 DEGREES
      Z=I
      A=(2.*Z-2.*DELAY*Z/(90.+DELAY))*RAD
      A=(COS(A)+1.)/2.
      SIA(J)=A
      SIA(J-2*I)=A
      I=I+1
35    CONTINUE
      DO 40 I=360,N
40    SIA(I-360)=SIA(I)
      N=180.-DELAY
      DO 45 I=N,180
45    SIA(I+360)=SIA(I)             ;END ANKLE TABLE WORK
C
      A=RAD*300./(TIME-140)
      XLEAN=TORSO*SIN(A)            ;UPPER BODY LEAN, F(SPEED), 2 DEG MODAL
      YLEAN=TORSO*COS(A)
      XINIT=130.                    ;INITIAL X VALUE IN SCREEN UNITS
      YINIT=550
      FALFO=0.                      ;MOVEMENT FALLING FORWARD W/IN OVERLAY PERIOD
                                    ;DUE TO TIBIA SWING AND FOOT
      DO 101 J=1,3                  ;3 STEP CYCLES (6 STEPS)
      DO 100 I=0,359,9              ;360 DEGREES PER CIRCLE IN 9 DEGREE INCREMENTS
      SINA=SI(I)                    ;MAKE SIN CALCULATIONS INTO STACK VARIABLES
      COSA=SI(I+90)
      COS2A=SI(I*2+90)
      SINA1=SIA(I)                  ;ANKLE ANGLES INTO STACK VARIABLES
      SINA2=SIA(I+180)
      IF(I.LE.180) COSI=COSA        ;ADVANCE DUE TO INVERSE PENDULAR MOTION
      IF(I.GT.180) COSI=SI(I-90)
      IF(I.GE.(180-DELAY).AND.I.LE.(180+DELAY) FALFO=FALFO+ STEP*.02
      IF(I.GE.(360-DELAY).OR.I.LE.DELAY) FALFO=FALFO+STEP*.02
                                    ;EXTRA X MOVEMENT DURING ANKLE FLEXION
      X=XINIT-COSI*STEP/2.+FALFO      ;MIDHIP X VALUE
      Y=YINIT-BOUNC*COS2A             ;MIDHIP Y VALUE
C
      XHEAD=X+1.6*XLEAN+TORSO/15.   ;HEAD LOCATION, W/ FORWARD PERCH,
                                    ;X AXIS
```

```
      YHEAD=Y+1.6*YLEAN                                          ;Y AXIS
      HX=HEX*COSA                 ;CALCULATIONS FOR HIPS
      HY=HEY*SINA-COSFF*COSA
      XHIPR=X+HX                  ;RIGHT HIP LOCATION, X AXIS
      YHIPR=Y+HY                                    ; Y AXIS
      XHIPL=X-HX                  ;LEFT HIP LOCATION, X AXIS
      YHIPL=Y-HY                                    ; Y AXIS
      A=SWF*COSA+FEMFO            ;CALCULATIONS FOR KNEES
      Z=SWF*COSA-FEMFO
      XKNER=XHIPR+FEMUR*SIN(A)  ;RIGHT KNEE LOCATION, X AXIS
      YKNER=YHIPR-FEMUR*COS(A)                        ; Y AXIS
      XKNEL=XHIPL-FEMUR*SIN(Z)  ;LEFT KNEE LOCATION, X AXIS
      YKNEL=YHIPL-FEMUR*COS(Z)                        ; Y AXIS
      HX=SEX*COSA                 ;CALCULATIONS FOR SHOULDERS
      HY=SEY*SINA
      XSHOR=X+XLEAN-HX            ;RIGHT SHOULDER LOCATION, X AXIS
      YSHOR=Y+YLEAN-HY                                    ; Y AXIS
      XSHOL=X+XLEAN+HX            ;LEFT SHOULDER LOCATION, X AXIS
      YSHOL=Y+YLEAN+HY                                    ; Y AXIS
      HX=HUMOR*SIN(SWH*COSA)  ;CALCULATIONS FOR ELBOWS
      HY=HUMOR*COS(SWH*COSA)
      XELBR=XSHOR-HX              ;RIGHT ELBOW LOCATION, X AXIS
      YELBR=YSHOR-HY                                  ; Y AXIS
      XELBL=XSHOL+HX              ;LEFT ELBOW LOCATION, X AXIS
      YELBL=YSHOL-HY                                  ; Y AXIS
      A=SWF*COSA+FEMFO-SWT*SINA1      ;CALCULATIONS FOR ANKLES
      Z=-SWF*COSA+FEMFO-SWT*SINA2
      XANKR=XKNER+TIBIA*SIN(A)   ;RIGHT ANKLE LOCATION, X AXIS
      YANKR=YKNER-TIBIA*COS(A)                          ; Y AXIS
      XANKL=XKNEL+TIBIA*SIN(Z)   ;LEFT ANKLE LOCATION, X AXIS
      YANKL=YKNEL-TIBIA*COS(Z)                          ; Y AXIS
      HX=SWH*COSA                 ;CALCULATIONS FOR WRIST
      HY=SWH*(COS2A+1.)/2.
      A=SWU*(COSA-1)/2.
      Z=SWU*(COSA+1.)/2.
      XWRIR=XELBR-ULNA*SIN(HX+A)      ;RIGHT WRIST LOCATION, X AXIS
      YWRIR=YELBR-ULNA*COS(HY-A)                             ; Y AXIS
      XWRIL=XELBL+ULNA*SIN(HX+Z)      ;LEFT WRIST LOCATION, X AXIS
      YWRIL=YELBL-ULNA*COS(HY+Z)                             ; Y AXIS
C                                 ;START FILLING DISPLAY TABLE
      N=IST                       ;INCREMENTS 2 WORDS PER CALL OF DDVA
                                  ;JOINTS SEEN EACH FRAME
      CALL DDVA(IFIX(XHEAD),IFIX(YHEAD),K,N)  ;X&Y VALUE, INTENSITY, POINTER
      CALL DDVA(IFIX(XKNER),IFIX(YKNER),K,N)
      CALL DDVA(IFIX(XANKR),IFIX(YANKR),K,N)
      CALL DDVA(IFIX(XSHOR),IFIX(YSHOR),K,N)
      CALL DDVA(IFIX(XELBR),IFIX(YELBR),K,N)
      CALL DDVA(IFIX(XWRIR),IFIX(YWRIR),K,N)
                                  ;JOINTS SEEN IN MANY BUT NOT ALL FRAMES
      IF(I.GT.110.AND.I.LT.130) GOTO 61
      CALL DDVA(IFIX(XANKL),IFIX(YANKL),K,N)
61    IF(I.GT.67.AND.I.LT.113.OR.I.GT.247.AND.I.LT.293) GOTO 62
      CALL DDVA(IFIX(XKNEL),IFIX(YKNEL),K,N)
62    IF(I.GT.48.AND.I.LT.95.OR.I.GT.265.AND.I.LT.312) GOTO 63
      CALL DDVA(IFIX(XHIPR),IFIX(YHIPR),K,N)
63    IF(I.GT.88.AND.I.LT.140.OR.I.GT.220.AND.I.LT.272) GOTO 64
      CALL DDVA(IFIX(XWRIL),IFIX(YWRIL),K,N)
64    IF(I.GT.30.AND.I.LT.155.OR.I.GT.215.AND.I.LT.330) GOTO 65
      CALL DDVA(IFIX(XELBL),IFIX(YELBL),K,N)
65    CONTINUE
      IST=IST+24                  ;(11 PTS + 1 STOP CODE) X 2 WORDS PER POINT
      IF(I-180) 100,80,100
80    XINIT=XINIT+STEP+FALFO
      FALFO=0.
100   CONTINUE                    ;END I LOOP, 1 STEP CYCLE
      XINIT=XINIT+STEP+FALFO
      FALFO=0.
101   CONTINUE                    ;END J LOOP, 3 CYCLES
C                                 ;DISPLAY WALKER SEQUENCE
```

```
150  IST=IP                           ;MINIMUM TIMES AT 0 & 180 DEGREES
     DO 200 I=0,119                   ;40 DISPLAYS PER CYCLE, 3 CYCLES
     J=TIME-SURGE*TIME*COS(I*18.*RAD)
     CALL SETPIT(J)                   ;DISPLAY TIME FOR UPCOMING FRAME, CLOCKED
     CALL DSTRT(IST)                  ;DISPLAY 1 FRAME
     IST=IST+24                       ;INCREMENT TO BEGINNING OF NEXT GRAME
199  IF(K3) 200, 199, 200             ;COUNT UNTIL DISPLAY TIME IS UP, FROM SETPIT
200  CONTINUE                         ;END DISPLAY LOOP
     CALL DSTOP                       ;CLEAR DISPLAY AT END OF SEQUENCE
     PAUSE                            ;HIT ANY KEY ON CONSOLE TO CONTINUE
     GOTO 150                         ;REDISPLAY SAME WALKER
     END
```

REFERENCES

Barclay, C. B., Cutting, J. E., & Kozlowski, L. T. Temporal and spatial factors in gait perception that influence gender recognition. *Perception & Psychophysics*, 1978, **23**, 145-152.

Carlsöö, S. *How man moves*. London: Heinemann, 1972.

Cooper, F. S., Liberman, A. M., & Borst, J. M. The interconversion of audible and visible patterns as a basis for research in the perception of speech. *Proceedings of the National Academy of Sciences*, 1951, **37**, 318-325.

Cutting, J. E., & Kozlowski, L. T. Recognizing friends by their walk: Gait perception without familiarity cues. *Bulletin of the Psychonomic Society*, 1977, **9**, 353-356.

Johansson, G. Visual perception of biological motion and a model for its analysis. *Perception & Psychophysics*, 1973, **14**, 201-211.

Johansson, G. Visual motion perception. *Scientific American*, 1975, **232** (6), 76-89.

Johansson, G. Spatio-temporal differentiation and integration in visual motion perception. *Psychological Research*, 1976, **38**, 379-393.

Kozlowski, L. T., & Cutting, J. E. Recognizing the sex of a walker from a dynamic point-light display. *Perception & Psychophysics*, 1977, **21**, 575-580.

Liberman, A. M., Cooper, F. S., Shankweiler, D. P., & Studdert-Kennedy, M. Perception of the speech code. *Psychological Review*, 1967, **74**, 431-461.

Marey, E. J. *Movement*. New York: Arno Press & New York Times, 1972. (Originally published in 1895).

(Accepted for publication November 7, 1977.)